


Descrierea și modelarea tranzistoarelor cu efect de câmp

Se va prezenta descrierea și modelarea în PSPICE a tranzistoarelor cu efect de câmp TEC-J și TEC-MOS.

Tranzistorul cu efect de câmp TEC-J

Forma generală de descriere este:

J<nume> <nod_drenă> <nod_poartă> <nod_sursă>
+<nume_model> [aria]



<nume_model> reprezintă numele modelului în care sunt descriși parametrii de model ai TEC-J. Comanda pentru definirea parametrilor de model ai TEC-J are forma următoare:

.MODEL <nume_model> NJF [parametri model] - pentru TEC-J cu canal N;
.MODEL <nume_model> PJF [parametri model] - pentru TEC-J cu canal P.


[aria] este un parametru opțional și reprezintă aria relativă a dispozitivului, având valoarea implicită 1.

Exemple:

J1 7 4 9 JMODN
J5 23 10 4 FETN 1.5

TEC-J este modelat ca un TEC intrinsec și rezistențe în serie cu drena ($RD/arie$) și cu sursa ($RS/arie$).

Modelul PSPICE al TEC intrinsec din figura alăturată conține diodele D_{GD} și D_{GS} , care modelează joncțiunile poartă-sursă și poartă-drenă, conductanțele G_{min} care ajută la convergența calculului PSF și sursa de curent I_D comandată de tensiunile V_{GS} și V_{DS} . Aceste elemente alcătuiesc modelul static al TEC-J. Modelul de semnal mare se obține dăugând capacitățile C_{GD} și C_{GS} la modelul static.



Modelul static al TEC-J

Pentru $V_{DS} > 0$, curentul de drenă este descris de ecuațiile:

$I_D = 0$, pentru $V_{GS} - V_{TO} \leq 0$ (regiunea de blocare);

$I_D = BETA(1 + LAMBDA \cdot V_{DS}) \cdot V_{DS} \cdot [2(V_{GS} - V_{TO}) - V_{DS}]$, pentru $V_{DS} \leq V_{GS} - V_{TO}$ (regiunea liniară);

$I_D = BETA(1 + LAMBDA \cdot V_{DS})(V_{GS} - V_{TO})^2$, pentru $0 < V_{GS} - V_{TO} < V_{DS}$ (regiunea de saturație).

Parametrii de model corespunzători modelului static al TEC-J sunt prezentați în tabelul următor.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
VTO	Tensiunea de prag	-2	volt
BETA	Coeficientul transconductanței	1E-4	amp/volt ²
LAMBDA	Parametrul de modulare a lungimii canalului de către tensiunea V_{DS}	0	volt ⁻¹
IS	Curentul de saturație al joncțiunilor porții	1E-14	amp
N	Coeficientul de emisie al joncțiunilor porții	1	
ISR	Parametrul curentului de recombinare al joncțiunilor porții	0	amp
NR	Coeficient de emisie al ISR	2	
RD	Rezistența drenei	0	ohm
RS	Rezistența sursei	0	ohm
ALPHA	Coeficientul de ionizare	0	volt ⁻¹
VK	Tensiunea de cot a ionizării	0	volt
XTI	Coeficientul de temperatură al IS	3	
VTOTC	Coeficient de temperatură al VTO	0	volt/°C
BETATCE	Coeficient exponențial de temperatură al BETA	0	%/°C

Modelul de semnal mare al TEC-J

Datorită sarcinilor stocate în cele două joncțiuni poartă-sursă și poartă-drenă, trecerea TEC-J din regimul de blocare în cel de conducție și invers nu se poate face instantaneu. Deoarece în funcționarea normală a TEC-J cele două joncțiuni sunt în permanență polarizate invers, sunt luate în considerație doar sarcinile statice ale ionilor necompensați din regiunile de tranziție ale acestor joncțiuni, iar aceste sarcini sunt modelate prin intermediul celor două capacități de barieră, a căror model este similar celui al diodelor. Parametrii utilizați în modelarea capacităților TEC-J sunt prezentați în tabelul de mai jos.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
CGS	Capacitatea de barieră poartă-sursă la polarizare nulă	0	farad
CGD	Capacitatea de barieră poartă-drenă la polarizare nulă	0	farad
M	Coeficient de gradare al joncțiunilor porții	0.5	
PB	Potențialul de barieră	1	V
FC	Coeficient al capacității de barieră la polarizare directă	0.5	


Zgomotul generat de TEC-J are două componente:

- zgomotul termic generat de rezistențele parazite RD și RS ;
- zgomotul de alicie și de licărire asociat sursei de curent I_D din TEC intrinsec.

Tranzistorul cu efect de câmp TEC-MOS

Forma generală de descriere a tranzistorului TEC-MOS este:

```
M<nume> <nod_drenă> <nod_poartă> <nos_sursă>
+<nod_substrat> <nume_model> [L=<valoare>] [W=<valoare>]
+[AD=<valoare>] [AS=<valoare>] [PD=<valoare>] [PS=<valoare>]
+[NRD=<valoare>] [NRS=<valoare>] [NRG=<valoare>]
+[NRB=<valoare>] [M=<valoare>]
```



<nume_model> reprezintă numele modelului în care sunt descriși parametrii de model ai TEC-MOS. Comanda pentru definirea parametrilor de model ai TEC-MOS are forma următoare:

```
.MODEL <nume_model> NMOS [parametri model] - pentru TEC-MOS cu canal N;
.MODEL <nume_model> PMOS [parametri model] - pentru TEC-MOS cu canal P.
```

Parametri opționali au semnificațiile următoare:

- L , W - lungimea, respectiv lățimea canalului. Ei pot fi specificați în cadrul dispozitivului, sau prin intermediul comenzilor .MODEL sau .OPTIONS. Acești parametri nu pot fi utilizați ca variabile de ieșire în analiza .MC;
- AD , AS - ariile de difuzie ale drenei, respectiv sursei;
- PD , PS - perimetrele zonelor de difuzie ale drenei, respectiv sursei;
- NRD , NRS , NRG , NRB - factorii de multiplicare (având unitatea de măsură pătratul) ai rezistenței pe pătrat (RSH), cu ajutorul cărora se determină rezistențele parazite ale drenei (RD), sursei (RS), porții (RG) și substratului (RB). Dacă se analizează o folie pătrată de material rezistiv, se constată că rezistența dintre două margini paralele ale acesteia depinde de compoziție și grosime, fiind independentă de dimensiunile sale atât timp cât este pătrată (de exemplu, latura pătratului poate fi 1 mm, 1 cm, 1 dm sau 1 m). Din acest motiv, rezistența foliei unui astfel de strat (RSH) are unitatea de măsură ohm/pătrat;
- M - factorul de multiplicare a dispozitivelor conectate în paralel (valoarea implicită fiind 1), care modelează efectul conectării în paralel a mai multor TEC-MOS. Astfel, capacitățile și curenții joncțiunilor vor fi multiplicare cu M , iar rezistențele parazite vor fi divizate cu M .

Exemple:

```
M1 3 5 12 0 MOSN M=2
```

```
M5 6 3 9 0 MODMOS L=5u W=2u
```

```
M30 15 7 6 0 NMOD L=12u W=6u AD=250p AS=250p PD=60u PS=60u
```

TEC-MOS este modelat ca un TEC-MOS intrinsec și rezistențe în serie cu drenea (RD), cu sursa (RS), cu poarta (RG) și cu substratul (RB). De asemenea, mai există o rezistență de șuntare în paralel cu canalul dintre drenă și sursă (RDS).

Simulatorul PSPICE dispune de șase modele de TEC-MOS, care diferă prin formularea caracteristicii curent-tensiune. Parametrul de model *LEVEL* definește modelul utilizat:

- $LEVEL=1$ este modelul Shichman-Hodges (acest model asigură o precizie scăzută);
- $LEVEL=2$ este un model analitic, descris de ecuațiile Meyer, bazat pe caracteristicile geometrice ale TEC-MOS, care asigură o precizie bună, însă poate determina apariția unor probleme de convergență;


- *LEVEL*=3 este un model semi-empiric, bazat pe ecuațiile Dang , care modelează efectele de canal scurt, fiind un model precis, dar care determină creșterea timpului de simulare;
- *LEVEL*=4 este modelul BSIM (Berkeley Short-Channel IGFET Model for MOS Transistor), în care valorile parametrilor sunt obținute din caracteristicile de proces;
- *LEVEL*=5 este modelul EKV versiunea 2.6;
- *LEVEL*=6 este modelul BSIM3 versiunea 2.0;
- *LEVEL*=7 este modelul BSIM3 versiunea 3.1.

Modelul TEC-MOS este prezentată în figura alăturată. Modelul static se obține eliminând capacitățile.

Parametrii de model comuni tuturor nivelelor de model sunt elementele parazite (rezistențe, capacități etc).

Caracteristicile statice ale primelor trei nivele de model sunt definite de parametrii *VTO*, *KP*, *LAMBDA*, *PHI* și *GAMMA*. Aceștia sunt calculați de simulator dacă se specifică parametrii de proces (de exemplu *TOX* și *NSUB*). În primele trei nivele de model, modelarea capacităților TEC-MOS a fost modificată față de celelalte nivele.

În tabelul de mai jos sunt prezentați parametrii nivelelor de model 1, 2 și 3.



Parametru model	Semnificație	Valoare implicită	Unitate măsură
LD	Lungimea difuziei laterale	0	metru
WD	Lățimea difuziei laterale	0	metru
VTO	Tensiunea de prag la polarizare nulă ($V_{BS}=0$)	0	volt
KP	Coeficientul transconductanței	2E-5	amp/volt ²
LAMBDA	Coeficientul de modulare a lungimii canalului de către tensiunea V_{DS} (nivelele 1, 2)	0	volt ⁻¹
PHI	Potențialul de suprafață al semiconductorului la inversiune	0.6	volt
GAMMA	Parametrul tensiunii de prag a substratului	Calculat de simulator	volt ^{1/2}
TOX	Grosimea stratului de oxid	Calculat de simulator	metru
TPG	Tipul de material al porții: +1 = opus substratului -1 = același cu substratul 0 = aluminiu	+1	
NSUB	Densitatea de dopare a substratului		1 / cm ³
XJ	Adâncimea joncțiunilor metalurgice (nivelele 2, 3)	0	metru
UO	Mobilitatea de suprafață a purtătorilor	600	cm ² / volt · sec

UCRIT	Câmpul critic al degradării mobilității (nivelul 2)	1E4	volt / cm
UEXP	Exponentul degradării mobilității (nivelul 2)	0	
VMAX	Viteza maximă de drift	0	metru/sec
NEFF	Coeficientul sarcinii din canal (nivelul 2)	1	
XQC	Fracțiunea din sarcina canalului atribuită drenei	1	
DELTA	Efectul lățimii canalului asupra tensiunii de prag	0	
THETA	Parametrul de modulare a mobilității de către câmpul din canal (nivelul 3)	0	volt ⁻¹
ETA	Coeficientul de reacție statică (nivelul 3)	0	
KAPPA	Factor al câmpului de saturație (nivelul 3)	0.2	

În tabelul următor sunt prezentați parametrii TEC-MOS comuni tuturor nivelelor de model.

Parametru model	Semnificație	Valoare implicită	Unitate măsură
LEVEL	Indicele modelului	1	
L	Lungimea canalului	DEFL	metru
W	Lățimea canalului	DEFW	metru
RD	Rezistența drenei	0	ohm
RS	Rezistența sursei	0	ohm
RG	Rezistența porții	0	ohm
RB	Rezistența substratului	0	ohm
RDS	Rezistența de șuntare drenă-sursă	infini	ohm
RSH	Rezistența pe pătrat de difuzie a drenei, sursei	0	ohm/pătrat
IS	Curentul de saturație al joncțiunii de substrat	1E-14	amp
JS	(Curentul de saturație a joncțiunii de substrat)/(arie)	0	amp/metru ²
JSSW	(Curentul lateral de saturație a joncțiunii de substrat)/(lungime)	0	amp/metru
N	Coeficientul de emisie al joncțiunii de substrat	1	
PB	Potențialul zonei inferioare a joncțiunii de substrat	0.8	volt
PBSW	Potențialul zonei laterale a joncțiunii de substrat	PB	volt
CBD	Capacitatea joncțiunii substrat-drenă la polarizare nulă	0	farad
CBS	Capacitatea joncțiunii substrat-sursă la polarizare nulă	0	farad
CJ	(Capacitatea zonei inferioare a joncțiunii de substrat la polarizare nulă)/(arie)	0	farad/metru ²
CJSW	(Capacitatea zonei laterale a joncțiunii de substrat la polarizare nulă)/(lungime)	0	farad/metru
MJ	Coeficientul de gradare a zonei inferioare a joncțiunii de substrat	0.5	
MJSW	Coeficientul de gradare a zonei laterale a joncțiunii de substrat	0.33	
FC	Coeficientul capacității joncțiunii de substrat la	0.5	

	polarizare directă		
TT	Timpul de tranzit al joncțiunii de substrat	0	sec
CGSO	(Capacitatea de suprapunere poartă-sursă)/(lățimea canalului)	0	farad/metru
CGDO	(Capacitatea de suprapunere poartă-drenă)/(lățimea canalului)	0	farad/metru
CGBO	(Capacitatea de suprapunere poartă-substrat)/(lungimea canalului)	0	farad/metru
KF	Coeficientul zgomotului de licărire	0	
AF	Exponentul zgomotului de licărire	1	

Ecuatiile curenților din modelul static TEC-MOS sunt următoarele:

Curentul de poartă: $I_g = 0$;

Curentul de substrat: $I_b = I_{bs} + I_{bd}$, unde:

$$I_{bs} = \text{curentul de scurgere substrat-sursă } I_{bs} = I_{ss} \cdot \left(e^{\frac{V_{bs}}{N \cdot V_t}} - 1 \right);$$

$$I_{bd} = \text{curentul de scurgere substrat-drenă } I_{bd} = I_{ds} \cdot \left(e^{\frac{V_{bd}}{N \cdot V_t}} - 1 \right), \text{ unde:}$$

- dacă $J_S=0$ sau $A_S=0$ sau $A_D=0$, atunci $I_{ss} = I_{ds} = I_S$,

- altfel, $I_{ss} = A_S \cdot J_S + P_S \cdot J_{SSW}$, iar $I_{ds} = A_D \cdot J_S + P_D \cdot J_{SSW}$

Curentul de drenă: $I_d = I_{drenă} - I_b$;

Curentul de sursă: $I_s = -I_{drenă} - I_b$, unde, pentru nivelul 1 de model și $V_{ds} > 0$:

- $I_{drenă}=0$, pentru $V_{gs} - V_{to} < 0$ (regiunea de blocare);

- $I_{drenă} = \frac{W}{L} \cdot \frac{K_P}{2} \cdot (1 + LAMBDA \cdot V_{ds}) \cdot V_{ds} \cdot [2 \cdot (V_{gs} - V_{to}) - V_{ds}]$, pentru $V_{ds} < V_{gs} - V_{to}$ (regiunea liniară);

- $I_{drenă} = \frac{W}{L} \cdot \frac{K_P}{2} \cdot (1 + LAMBDA \cdot V_{ds}) \cdot (V_{gs} - V_{to})^2$, pentru $0 \leq V_{gs} - V_{to} \leq V_{ds}$ (regiunea de saturație), unde $V_{to} = V_{TO} + GAMMA \cdot (\sqrt{PHI - V_{bs}} - \sqrt{PHI})$.

Aplicația 1: Caracteristicile statice ale TEC-J

J1 2 1 0 J2N3369

.MODEL J2N3369 NJF(Beta=336.2u Betatce=-.5 Rd=1 Rs=1 Lambda=11m Vto=-2.07 +Vtotc=-2.5m Is=114.5f Isr=1.091p N=1 Nr=2 Xti=3 Alpha=506.8u Vk=251.7 Cgd=2.8p +M=.2271 Pb=.5 Fc=.5 Cgs=2.916p Kf=1.214E-18 Af=1)

VGS 1 0 DC -2

VDS 2 0 DC 5

*Caracteristica de transfer $I_d(V_{gs})$ pentru diferite valori ale V_{ds} , respectiv ale temperaturii

.DC VGS -3 0 0.05

.STEP VDS LIST 1 2.5 5

*.TEMP -25 25 75

*Caracteristica de iesire $I_d(V_{ds})$ pentru diferite valori ale V_{gs}

*.DC DEC VDS 1m 5 50

*.STEP VGS LIST -2.25 -1.75 -1.25 -0.75

.PROBE

.END


Aplicația 2: Caracteristicile statice ale TEC-MOS.

```
M1 2 1 0 0 CMOSN L=1u W=1u
.MODEL CMOSN NMOS(Level=3 Phi=0.60 Tox=2.03E-8 Xj=0.15U Tpg=1 Vto=0.7333
+Delta=9.445E-1 Ld=1E-9 Kp=1.2964E-4 Uo=762.1 Theta=5.246E-2 Rsh=2.365
+Gamma=0.4481 Nsub=1.75E+16 Nfs=2.356E+12 Vmax=1.487E+5 Eta=1.485E-1
+Kappa=9.51E-2 Cgdo=2.5516E-12 Cgso=2.5516E-12 Cgbo=3.0108E-10 Cj=1.1962E-4
+Mj=0.4398 Cjsw=4.6935E-10 Mjsw=0.123994 Pb=0.8)
VDS 2 0 DC 3
VGS 1 0 DC 1
*Caracteristica de transfer Id(Vgs) pentru diferite valori ale Vds, respectiv ale temperaturii
.DC VGS 0 3 50m
.STEP VDS LIST 1 2 3
*.TEMP -25 25 75
*Caracteristica de iesire Id(Vds) pentru diferite valori ale Vgs
*.DC VDS -0.5 5 0.1
*.STEP VGS LIST 0.5 1 1.5
.PROBE
.END
```

Aplicația 3: Circuit cu 2 inversoare logice realizate cu TEC-MOS complementare (CMOS)

Inversorul logic va fi definit ca subcircuit și va fi apelat de două ori. M1 și M2 utilizează modelele CMOSN, respectiv CMOSP, salvate în fișierul CMOS08.LIB.

```
*Descrierea circuitului
VIN 1 0 PULSE(0 5 5n 1n 1n 10n 25n)
XINV1 1 10 2 INVMOS
XINV2 2 10 3 INVMOS
CL1 2 0 0.3p
CL2 3 0 0.3p
VDD 10 0 DC 5
*1=IN, 3=VDD, 2=IES
.SUBCKT INVMOS 1 3 2
M1 2 1 0 0 CMOSN L=1u W=2u
M2 2 1 3 3 CMOSP L=1u W=2u
.ENDS
.LIB "D:\LIB\CMOS08.LIB"
.TRAN 0.5n 25n 0 0.2n
.DC VIN 0 5 50m
.PROBE V(1) V(2) V(3)
.END
```




În urma analizei de regim tranzitoriu se vor reprezenta grafic tensiunile nodurilor 1, 2 și 3. Să se determine timpii de întârziere ai formelor de undă de la ieșirea primului inversor.


În urma analizei de curent continuu se va reprezenta grafic caracteristica de transfer statică a primului inversor logic (V(2)).

Aplicația 4: Poartă logică NOR cu 2 intrări, realizată cu TEC-MOS complementare (CMOS)

MN1, MN2 și MP1, MP2 utilizează modelele CMOSN, respectiv CMOSP, salvate în fișierul CMOS08.LIB, iar lungimea și lățimea canalului au valorile $L=1\mu$, respectiv $W=2\mu$.



La prima intrare a circuitului se conectează sursa de tensiune V1 cu specificație de curent continuu de valoare 0V sau 5V, iar la a doua intrare se conectează sursa de tensiune V2 cu specificația tranzitorie din figura de mai jos.



Se vor efectua o analiză în domeniul timp și o analiză de curent continuu:

.TRAN 0.1n 20n 0 0.05n

.DC V2 0 5 50m.

În urma analizei .TRAN reprezentați grafic tensiunile de la intrarea și ieșirea circuitului, precum și curentul absorbit din sursa de alimentare.

În urma analizei .DC reprezentați grafic caracteristica de transfer statică a porții logice ($V(3)$) și curentul absorbit din sursa de alimentare ($-I(VDD)$).